Grace version

Download gearFactory.grace Delete

```
1  factory method gearFromChainring(ch) cog(cg) rim(r) tire(t) {
2      def chainring is public = ch
3      def cog is public = cg
4      method tire { t }
5      method rim { r }
6      method ratio {
7        chainring / cog
8      }
9      method gearInches {
10       (ratio * (rim + (tire * 2))*10).rounded/10
11     }
12 }
13
14 def g1 = gearFromChainring 52 cog 11 rim 26 tire 1.5
15 print "a {g1.chainring}-T chainring and {g1.cog}-T cog on a {g1.rim}-inch rim provides a {g1.gearInches} inch gear"
16 def g2 = gearFromChainring 52 cog 11 rim 24 tire 1.25
17 print "a {g2.chainring}-T chainring and {g2.cog}-T cog on a {g2.rim}-inch rim provides a {g2.gearInches} inch gear"
18 print "g1 is {g1}"
```

Build

```
a 52-T chainring and 11-T cog on a 26-inch rim provides a 137.1 inch gear
a 52-T chainring and 11-T cog on a 24-inch rim provides a 125.3 inch gear
g1 is an object
```

Single Responsibility?

- Not really!
  - since when does a gear have a a *tire* and a *rim*?
  - mixed up with other bits of *bicycle*
- Does it matter?
  - *Maybe!*

Arguments to “Leave it be”

- Code is *Transparent* and *Reasonable*
  - Consequence of a change should be *Transparent*
  - Cost of change *proportional* to benefits
- Why? Because there *are no dependents*
- *How* should we improve it?
  - We don't yet know — but the moment that we *acquire some dependents*, we will

➡ Wait until that time

Arguments for Change

- Code is neither (re)usable, nor exemplary
  - multiple responsibilities ⇒ can't be used for other *gear*s
  - not a pattern to be emulated

Improve it now *vs.* improve it later

- This tension always exists!
  - designs are never perfect
  - the designer has to weigh the costs and benefits of a change

Embracing Change

- Some basic rules that will help, regardless of what change happens:
- Depend on Behaviour, not data
  - encapsulate instance variables
    - Grace gives us this one for free
  - encapsulate data
    - e.g., don’t expose an array of pairs of numbers

method knows all about the structure of *d*

Download obscringReferences.g

```
1  factory method obscuringReferences(d) {
2      method diameters {
3        data.map { pair -> pair.first + (pair.second * 2) }
4      }
5      def data is public = d
6  }
7
8  def or = obscuringReferences(
9    list.with(
10     list.with(622, 20), list.with(622, 23), list.with(559, 30), list.with (559, 40)
11   )
12 )
13
14 print(or.diameters.asList)
15 print(or.data)
16
```

Run ▶

```
[662,668,619,639]
[[622,20],[622,23],[559,30],[559,40]]
```

Separate Structure from Meaning

- If you need a table of wheel and tire sizes, make it contain *objects*, not lists
- Metz uses a Ruby Struct to create a transparent object.
- In Grace:

```
factory method wheelWithRim(r) tire(t) {
    // this is equivalent to the Ruby `Struct.new(:rim, :tire)`
    method rim { r }
    method tire { t }
    method asString { "{rim} wheel with {tire} tire" }
}
```

Download revealingReferences.grace Delete

```
factory method revealingReferences(d:List<List<Number>>) {
    method diameters {
      wheels.map { each -> each.rim + (each.tire * 2) }
    }
    def wheels is public = wheelify(d)
    method wheelify(pairs) {
      pairs.map { pair -> wheelWithRim(pair.first) tire(pair.second) }.asList
    }
    factory method wheelWithRim(r) tire(t) {
        // this is equivalent to the Ruby `Struct.new(:rim, :tire)`
        method rim { r }
        method tire { t }
        method asString { "{rim} wheel with {tire} tire" }
    }
}



def rr = revealingReferences(
  list.with(
    list.with(622, 20), list.with(622, 23), list.with(559, 30), list.with (559, 40)
  )
)

print(rr.diameters.asList)
print(rr.wheels)

```

Run ▶

```
[662,668,619,639]
[622 wheel with 20 tire,622 wheel with 23 tire,559 wheel with 30 tire,559 wheel with 40 tire]
```

Embracing Change

- Enforce Single Responsibility Everywhere
  - Extract extra responsibilities from methods
  - Isolate responsibilities in classes
    - Grace lets you create “local” factory methods

The Real Wheel

- The customer tells you that she has need for computing wheel circumference.
- This tells you that your “bicycle calculator app” needs to model wheels.
- So let’s move *wheel* out of *gear*.